Physics through Computational Thinking

Driven oscillations

Auditya Sharma and Ambar Jain
Dept. of Physics, IISER Bhopal

Outline

In this module we will look at

The improved Euler method and how it can dramatically improve over the Euler method.

The Driven Oscillator.

```
Clear["Global`*"]
```

We have learned some very simple numerical techniques for solving differential equations. So far we have only used these methods to study very simple problems which we independently know how to solve analytically. Next we would like to set up more difficult problems where analytical methods become harder and harder and test these numerical methods. We also wish to develop the numerical methods further so that they can be made as efficient as possible. Eventually once we have enough faith in our numerics we would want to tackle problems that are intractable analytically.

One problem of great interest is the driven oscillator. Suppose we have a harmonic oscillator set up with a mass m attached to a spring of spring constant k. The equation of motion is the familiar:

$$m \frac{d^2 x}{d t^2} = -k\, x. \tag{1}$$

This would of course give simple harmonic motion with the natural frequency $\omega_0 = \sqrt{\frac{k}{m}}$. Suppose in addition we have an external periodic force that drives the mass at a frequency ω. This would correspond to a differential equation of the type:

$$m \frac{d^2 x}{d t^2} = -k\, x + F \cos(\omega\, t). \tag{2}$$

Exercise

(a) Non-dimensionalize the equation by choosing suitable scales expressing the equation in dimensionless quantities.

(b) How many free parameters are left in the equation after non-dimensionalization?

Solution

$$\omega \text{ scale}: \quad \omega_0 = \sqrt{\frac{k}{m}}$$

$$\begin{aligned} t\ \text{scale}: &\quad \frac{1}{\omega_0} = \sqrt{\frac{m}{k}} \\ a\ \text{scale}: &\quad \frac{F}{m} \\ x\ \text{scale}: &\quad \sqrt{a\, t^2} = \frac{F}{k} \end{aligned}$$

Making the transformation:

$$\begin{aligned} x &\longrightarrow \frac{F}{k}\, x \\ t &\longrightarrow \frac{1}{\omega_0}\, t \\ \omega &\longrightarrow \omega_0\, \omega \end{aligned} \tag{4}$$

we get

$$\begin{aligned} m \frac{F}{k} \omega_0{}^2 \frac{d^2 x}{d t^2} &= -k \frac{F}{k} x + F \cos(\omega\, t) \\ \Rightarrow \frac{d^2 x}{d t^2} &= -x + \cos(\omega\, t) \end{aligned} \tag{5}$$

After non-dimensionalization, there is only one free parameter namely the driving frequency ω which is left in the problem.

Let us assume that the initial conditions for this problem in dimensionless units is given by $x(0)=1$ and $\dot{x}(0) = 0$.

This is a second order differential equation which can be solved exactly analytically for all times. However, it is instructive to just focus on the steady state behavior of this system. For long times, the motion of the particle would be entirely dominated by the driving frequency and it is natural to guess that the system simply oscillates with the driving frequency. So for the steady state we can make the educated guess (sometimes called an ansatz):

$$x_{\text{ss}}(t) = C \cos(\omega\, t), \tag{6}$$

where the constant C needs to be determined.

Exercise

(c) Implant the ansatz into the differential equation and work out $C(\omega)$

(d) Plot $|C(\omega)|^2$. Explain what it means.

Solution

If we implant the ansatz into the differential equation we have

$$-C\,\omega^2 \cos(\omega\, t) = -C \cos(\omega\, t) \;+\; \cos(\omega\, t), \tag{7}$$

which yields

$$C\,(\omega) = \frac{1}{1-\;\omega^2}. \tag{8}$$

A plot of the square of the amplitude as a function of ω is instructive. It tells how the strength of the oscillations would be in the steady state depending on the frequency with which the system is driven.

```
Csq[ω_] = 1/(1 - ω^2)^2;
Plot[Csq[ω], {ω, 0, 2}, PlotRange → Automatic, AxesLabel → {ω, C^2}];
```

We see that if the system is driven at a resonant frequency the amplitudes can become singular. This is a familiar concept and finds wide-ranging applications. In some contexts it is desirable to drive it at resonance so that large amplitudes may be accomplished. But on the other hand there are contexts where resonance must be avoided, in order to prevent damage. For instance, armies marching on bridges are advised to break step so that they do not inadvertently drive the bridge at its natural frequency which could result in devastating consequences!

Full Solution of the Driven Oscillator.

The steady state solution we have obtained is surely not the full solution of the differential equation, because all the information about the initial conditions is missing! In fact, the steady state solution better not depend on initial conditions, because steady state is a long-time phenomenon and the system should get to there regardless of where it started. Also a second order differential equation must have to two free constants, which are fixed with the help of the initial conditions. This information plays out crucially when we consider the transient behavior of the system.

What we have already found as a steady state solution is called a particular solution. The theory says that the full general solution is obtained by simply adding to this particular solution what is called the complementary solution of the corresponding homogeneous differential equation, which in this case is simply

$$\frac{d^2 x}{d t^2} = -x \tag{9}$$

Exercise

(a) Find the general complementary solution $x_c(t)$ of the homogeneous differential equation above. How many free constants does it contain?

(b) Now write down the full general solution of the problem as $x(t) = x_c(t) + x_{\text{ss}}(t)$. Check that this solution works explicitly by plugging into the original differential equation.

(c) Now plug in the initial conditions to fix the free constants.

(d) Plot the solution for a range of the driving frequency ω. Discuss the solution.

(e) What about resonant driving? What is the solution to this problem? Take the limit appropriately to extract the solution at this point.

Solution

(a) The complementary solution is of course well known:

$$x_c(t) = c_1 \cos(t) + c_2 \sin(t) \tag{10}$$

Since it the general solution of a second-order differential equation it better have two free constants, and it does.

(b) We are now ready to write down the full general solution of the original differential equation:

$$x(t) = c_1 \cos(t) + c_2 \sin(t) + \frac{1}{1-\omega^2} \cos(\omega\, t) \tag{11}$$

(c) Now plugging in the initial conditions we have :

$$x(0) = c_1 + \frac{1}{1-\omega^2} = 1$$
$$\dot{x}(0) = c_2 = 0.$$
$$\Rightarrow x(t) = \frac{1}{1-\omega^2}\left[-\omega^2 \cos(t) + \cos(\omega\, t)\right]$$

which is the full solution of the problem.

(d) Now we plot this solution :

```
Manipulate[Plot[(-ω^2 Cos[t] + Cos[ω t])/(1 - ω^2), {t, 0, 1000}, PlotLabel → ω], {ω, 0, 2}];
```

What we are seeing is nothing but beats since the solution is just superposing two cosine functions! The beats are particularly evident when the two frequencies are close to each other, that is close to resonance.

It turns out that the problem at resonance requires a special handling and the solution exactly at that point is given by

$$x(t) = \lim_{\omega\to 1} \frac{1}{1-\omega^2}\left[-\omega^2 \cos(t) + \cos(\omega\, t)\right]$$
$$\Rightarrow x(t) = \cos(t) + \frac{t}{2}\sin(t) \qquad (13)$$

```
Plot[Cos[t] + t/2 Sin[ t], {t, 0, 100}];
```

As expected, the amplitude of the vibrations become larger and larger without bound. The above limit, by the way, could have been evaluated with the help of *Mathematica:*

```
Limit[(-ω^2 Cos[t] + Cos[ω t])/(1 - ω^2), ω → 1];
```

```
Cos[t] + 1/2 t Sin[t]
```

This is exactly what we already obtained analytically. Such symbolic calculations make *Mathematica* a powerful tool!

Improved Euler's Method

- ***Improved Euler's method*** improves the Euler method by reducing the local error to order h^3 and global error to order h^2.
- This is how Improved Euler Method is defined:

$$\begin{aligned} t_{n+1} &= t_n + h \\ \tilde{x}_{n+1} &= x_n + h\, f(t_n, x_n) \\ x_{n+1} &= x_n + h\, \frac{f(t_n, x_n) + f(t_{n+1}, \tilde{x}_{n+1})}{2} \end{aligned} \tag{14}$$

- This can also be written as

$$\begin{aligned} t_{n+1} &= t_n + h \\ x_{n+1} &= x_n + h\, \frac{f(t_n, x_n) + f(t_n + h,\ x_n + h\, f(t_n, x_n))}{2} \end{aligned} \tag{15}$$

- The improved Euler method is also known as the second-order Runge Kutta (RK) method.

Exercise

(a) Improve the **eulerGen** function to obtain a code that implements Improved Euler Method. Call this function as **eulerImp**.

(b) Solve the driven harmonic oscillator equation discussed with **eulerGen** and **eulerImp** methods. Compare the accuracy of your computation in each of the cases. See what is nMax that you require in each case to obtain an accurate result.

Solution

⋮

```
eulerImp[F_, X0_, tf_, nMax_] := Module[{h, datalist, prev, next1, next, rate, rate1},
  h = (tf - X0[[1]]) / nMax // N;
  For[datalist = {X0},
   Length[datalist] ≤ nMax,
   AppendTo[datalist, next],
   prev = Last[datalist];
   rate = Through[F @@ prev];
   next1 = prev + h rate;
   rate1 = Through[F @@ next1];
   next = prev + h/2 (rate + rate1);
  ];
  Return[datalist];
 ]

Clear[ω]
Id[t_, x_, v_] = 1;
xDot[t_, x_, v_] = v;
vDot[t_, x_, v_] = -x + Cos[ω t];
initial = {0, 1, 0};

ω = 0.9;
data = eulerGen[{Id, xDot, vDot}, initial, 100, 10000];
Show[ListPlot[data[[ ;; , 1 ;; 2]], Joined → True, PlotMarkers → None, PlotRange → Full],
  Plot[(-ω^2 Cos[t] + Cos[ω t]) / (1 - ω^2), {t, 0, 100}, PlotRange → Full, PlotStyle → Red]];

ω = 0.9;
data = eulerImp[{Id, xDot, vDot}, initial, 100, 2000];
Show[ListPlot[data[[ ;; , 1 ;; 2]], Joined → True, PlotMarkers → None, PlotRange → Full],
  Plot[(-ω^2 Cos[t] + Cos[ω t]) / (1 - ω^2), {t, 0, 100}, PlotRange → Full, PlotStyle → Red]];
```

The Resonant Case:

```
ω = 1;
data = eulerGen[{Id, xDot, vDot}, initial, 100, 10 000];
Show[ListPlot[data[[ ;; , 1 ;; 2]], Joined → True, PlotMarkers → None, PlotRange → Full],
  Plot[Cos[t] + t/2 Sin[ t], {t, 0, 100}, PlotRange → Full, PlotStyle → Red]];
```

```
ω = 1;
data = eulerImp[{Id, xDot, vDot}, initial, 100, 1000];
Show[ListPlot[data[[ ;; , 1 ;; 2]], Joined → True, PlotMarkers → None, PlotRange → Full],
  Plot[Cos[t] + t/2 Sin[ t], {t, 0, 100}, PlotRange → Full, PlotStyle → Red]];
```

Some fun and games

In this exercise, we use the *Mathematica* function called Play to *listen* to the sounds corresponding to various harmonic wave forms.

```
Clear["Global`*"]

A = 100;

ω1 = 2 π 261.63;
ω2 = 2 π 293.66;
ω3 = 2 π 329.63;
ω4 = 2 π 349.23;
ω5 = 2 π 392;
ω6 = 2 π 440;
ω7 = 2 π 493.88;
ω8 = 2 π 523.25;

f1[t_] = A Cos[ω1 t];
f2[t_] = A Cos[ω2 t];
f3[t_] = A Cos[ω3 t];
f4[t_] = A Cos[ω4 t];
f5[t_] = A Cos[ω5 t];
f6[t_] = A Cos[ω6 t];
f7[t_] = A Cos[ω7 t];
f8[t_] = A Cos[ω8 t];

Play[f1[t], {t, 0, 2}]
Play[f2[t], {t, 0, 2}]
Play[f3[t], {t, 0, 2}]
Play[f4[t], {t, 0, 2}]
Play[f5[t], {t, 0, 2}]
Play[f6[t], {t, 0, 2}]
Play[f7[t], {t, 0, 2}]
Play[f8[t], {t, 0, 2}]
```


2 s
8000 Hz


2 s
8000 Hz


2 s
8000 Hz


2 s
8000 Hz


2 s
8000 Hz







$\omega_A = 2\,\pi\;\;10\,440;$
$\omega_B = 2\,\pi\;\;10\,000;$

```
f[t_] = A Cos[ω_A t];
g[t_] = A Cos[ω_B t];
h[t_] = f[t] + g[t];
Play[{h[t]}, {t, 0, 2}]
```