Physics through Computational Thinking

Escape velocity

Auditya Sharma and Ambar Jain
Dept. of Physics, IISER Bhopal

Outline

In this module we will look at
differential equations that pertain to rockets escaping the earth's atmosphere.

Escaping the Earth

```
Clear["Global`*"]
```

We are accustomed to treating the acceleration due to gravity as a constant. This is true only if the body in question is very close to the Earth's surface. According to Newton's law of gravitation, an inverse square law force applies. Therefore, if x is the distance from the centre of the Earth to the projectile, the differential equation is

$$m \frac{d^2 x}{d t^2} = -\frac{G M m}{x^2}, \tag{1}$$

where M and m are the masses respectively of the Earth, and the projectile, and G is the universal gravitational constant. If R is the radius of the Earth, we have $g = \frac{G M}{R^2}$, and therefore the equation becomes

$$\frac{d^2 x}{d t^2} = -\frac{g R^2}{x^2}. \tag{2}$$

Defining the velocity $v = \frac{dx}{dt}$, we have

$$\frac{d^2 x}{d t^2} = \frac{d v}{d t} = \frac{d v}{d x} \frac{d x}{d t} = v \frac{d v}{d x}. \tag{3}$$

Therefore, our differential equation can be recast into

$$v \frac{d v}{d x} = -\frac{g R^2}{x^2}, \tag{4}$$

which after rearrangement becomes

$$v\, d v = -\frac{g R^2}{x^2} d x. \tag{5}$$

Integrating, we have

$$\frac{1}{2} v^2 = \frac{g R^2}{x} + c \tag{6}$$

If the speed with which the projectile is launched at the surface of the Earth is v_0, we have $c = \frac{1}{2} v_0{}^2 - g R$, and thus

$$v^2 = v_0{}^2 - \frac{2 g R(x - R)}{x} . \tag{7}$$

If the projectile must critically escape the Earth's gravitation, it means that $v \to 0$, as $x \to \infty$. Thus, the minimum speed with which the projectile must be hurled so that it escapes the Earth is:

$$v_0 = \sqrt{2 g R} . \tag{8}$$

Nondimensionalization

Let us rework this problem in a manner that can be tested numerically. We start by rewriting the differential equation as:

$$\frac{d^2 x}{d t^2} = -\frac{g R^2}{x^2}. \tag{9}$$

Exercise

(a) Non-dimensionalize the equation by choosing suitable scales expressing the equation in dimensionless quantities.

(b) How many free parameters are left in the equation after non-dimensionalization?

Solution

$$\begin{array}{rl} a \text{ scale}: & g \\ x \text{ scale}: & R \\ t \text{ scale}: & \sqrt{\frac{R}{g}} \end{array} \tag{10}$$

Making the transformation:

$$\begin{array}{ccc} x & \longrightarrow & R\, x \\ t & \longrightarrow & \sqrt{\frac{R}{g}}\; t \end{array} \tag{11}$$

we get

$$\frac{R}{\frac{R}{g}} \frac{d^2 x}{d t^2} = -\frac{g R^2}{R^2 x^2}. \tag{12}$$

$$\Rightarrow \frac{d^2 x}{d t^2} = -\frac{1}{x^2}$$

After non-dimensionalization, there is *no* free parameter left in the problem!

Let us assume that the initial conditions for this problem in dimensionless units is given by $x(0) = 1$ and $\dot{x}(0) = v_0$.

This is a second order differential equation which can be solved exactly. Defining $v = \frac{dx}{dt}$, we have

$$v \frac{d v}{d x} = -\frac{1}{x^2}. \tag{13}$$

Integrating, we have

$$v = \sqrt{\left(v_0{}^2 - 2\right) + \frac{2}{x}}\,. \tag{14}$$

A plot of this function is very instructive.

```
Manipulate[Plot[Sqrt[(v0^2 - 2) + 2/x], {x, 0, 1000}, PlotLabel → v0, AxesLabel → {x, v}], {v0, 0, 2}];
```

We are unaware of a simple closed-form solution for $x(t)$ for arbitrary v_0. However, if $v_0 = \sqrt{2}$, the critical value that allows the particle to escape to infinity, the integration is possible exactly, and we have for this case

$$x(t) = \left(\frac{3}{2}\sqrt{2}\, t + 1\right)^{2/3} \tag{15}$$

Plotting this function we have

```
xfunc[t_] = (3/2 √2 t + 1)^(2.0/3.0);
Plot[xfunc[t], {t, 0, 4000}, PlotRange → Automatic, AxesLabel → {t, x}];
```

Numerical Solution with the RK4 Method

- Lets recall how we can bring a higher order differential equation into the canonical form:

$$\begin{aligned}\dot{x} &= f(t, x, y, z)\\ \dot{y} &= g(t, x, y, z)\\ \dot{z} &= h(t, x, y, z)\end{aligned} \tag{16}$$

- Next we define the column vectors X and F as

$$X = \begin{pmatrix} t \\ x \\ y \\ z \end{pmatrix} \qquad F = \begin{pmatrix} 1 \\ f \\ g \\ h \end{pmatrix} \tag{17}$$

- Then the coupled ODEs can be written as

$$\dot{X} = F \tag{18}$$

- The RK4 method is given by

$$\begin{aligned}R_1 &= F(X_n)\\ R_2 &= F\left(X_n + \frac{h}{2} R_1\right)\\ R_3 &= F\left(X_n + \frac{h}{2} R_2\right)\\ R_4 &= F(X_n + h\, R_3)\end{aligned} \tag{19}$$

$$X_{n+1} = X_n + h\, \frac{R_1 + 2\, R_2 + 2\, R_3 + R_4}{6} \tag{20}$$

- Here we have copied its implementation.

```
rk4[F_, X0_, tf_, nMax_] := Module[{h, datalist, prev, rate1, rate2, rate3, rate4, next},
  h = (tf - X0[[1]]) / nMax // N;
  For[datalist = {X0},
   Length[datalist] ≤ nMax,
   AppendTo[datalist, next],
   prev = Last[datalist];
   rate1 = F@prev;
   rate2 = F@(prev + h/2 rate1);
   rate3 = F@(prev + h/2 rate2);
   rate4 = F@(prev + h rate3);
   next = prev + h/6 (rate1 + 2 rate2 + 2 rate3 + rate4);
  ];
  Return[datalist];
]
```

- The differential equation corresponding to the falling body subject to air resistance can be recast into canonical form as:

$$\begin{gathered} \frac{dx}{dt} = v \\ \frac{dv}{dt} = \frac{-1}{x^2} \\ x(0) = 1 \\ v(0) = v_0 \end{gathered} \tag{21}$$

- So in vector form we have:

$$X = \begin{pmatrix} t \\ x \\ v \end{pmatrix} \qquad F = \begin{pmatrix} 1 \\ v \\ \frac{-1}{x^2} \end{pmatrix} \qquad \dot{X} = F \tag{22}$$

- So we proceed to define the functions and the initial vector:

```
rateFunc[{t_, x_, v_}] = {1, v, -1/x^2};
initial = {0, 1, Sqrt[2]};
solx[t_] = (3/2 Sqrt[2] t + 1)^(2.0/3.0);
```

$$\texttt{rateFunc[\{t\_, x\_, v\_\}]} = \left\{1, v, \frac{-1}{x^2}\right\};$$

$$\texttt{initial} = \left\{0, 1, \sqrt{2}\right\};$$

$$\texttt{solx[t\_]} = \left(\frac{3}{2}\sqrt{2}\, t + 1\right)^{2.0/3.0};$$

- Now we are ready to invoke the rk4 function:

```
data = rk4[rateFunc, initial, 4, 300];

ListPlot[data[[ ;; , 1 ;; 2]], Joined -> True, PlotMarkers -> None, PlotRange -> Full];

Show[ListPlot[data[[ ;; , 1 ;; 2]], Joined -> True, PlotMarkers -> None, PlotRange -> Full],
  Plot[solx[t], {t, 0, 4}, PlotRange -> Full, PlotStyle -> Red]];
```

Taking into account air resistance

$$m \frac{d^2 x}{dt^2} = -\frac{m g R^2}{x^2} - k e^{-\lambda(x-R)} \frac{dx}{dt}. \tag{23}$$

Exercise

(a) Non-dimensionalize the equation by choosing suitable scales expressing the equation in dimensionless quantities.

(b) How many free parameters are left in the equation after non-dimensionalization?

Solution

$$\begin{array}{rl} a\text{ scale}: & g \\ x\text{ scale}: & R \\ t\text{ scale}: & \sqrt{\frac{R}{g}} \end{array} \tag{24}$$

Making the transformation:

$$\begin{array}{rcl} x & \longrightarrow & R x \\ t & \longrightarrow & \sqrt{\frac{R}{g}}\, t \end{array} \tag{25}$$

we get

$$\frac{R}{\frac{R}{g}} m \frac{d^2 x}{dt^2} = -\frac{m g R^2}{R^2 x^2} - k \frac{R}{\sqrt{\frac{R}{g}}} e^{-\lambda(x-R)} \frac{dx}{dt}. \tag{26}$$

$$\Rightarrow \frac{d^2 x}{d t^2} = -\frac{1}{x^2} - \frac{k}{m}\sqrt{\frac{R}{g}}\, e^{-\lambda R(x-1)} \frac{d x}{d t}$$

Now after non-dimensionalization, there are *two* free parameters left in the problem. Let us define two dimensionless free parameters $\alpha = \frac{k}{m}\sqrt{\frac{R}{g}}$, and $\beta = \lambda$ R, we have the non-dimensionalized equation

$$\frac{d^2 x}{d t^2} = -\frac{1}{x^2} - \alpha\, e^{-\beta(x-1)} \frac{d x}{d t} \tag{27}$$

Let us assume that the initial conditions for this problem in dimensionless units is given by $x(0) = 1$ and $\dot{x}(0) = v_0$.

This is a second order differential equation which cannot be solved exactly.

Numerical Solution with the RK4 Method

```
rk4[F_, X0_, tf_, nMax_] := Module[{h, datalist, prev, rate1, rate2, rate3, rate4, next},
  h = (tf - X0[[1]]) / nMax // N;
  For[datalist = {X0},
   Length[datalist] ≤ nMax,
   AppendTo[datalist, next],
   prev = Last[datalist];
   rate1 = F@prev;
   rate2 = F@(prev + h/2 rate1);
   rate3 = F@(prev + h/2 rate2);
   rate4 = F@(prev + h rate3);
   next = prev + h/6 (rate1 + 2 rate2 + 2 rate3 + rate4);
  ];
  Return[datalist];
 ]
```

- The differential equation corresponding to the falling body subject to air resistance can be recast into canonical form as:

$$\begin{gathered}\frac{dx}{dt} = v\\ \frac{dv}{dt} = -\frac{1}{x^2} - \alpha\, e^{-\beta(x-1)}\, v\\ x(0) = 1\\ v(0) = v_0\end{gathered} \tag{28}$$

- So in vector form we have:

$$X = \begin{pmatrix} t \\ x \\ v \end{pmatrix} \qquad F = \begin{pmatrix} 1 \\ v \\ -\frac{1}{x^2} - \alpha\, e^{-\beta(x-1)}\, v \end{pmatrix}$$

$$\dot{X} = F$$

- So we proceed to define the functions and the initial vector:

```
rateFunc[{t_, x_, v_}] = {1, v, -1/x^2 - ⅇ^(-(x-1)) v};
initial = {0, 1, 2.1};
```

- Now we are ready to invoke the rk4 function:

```
data = rk4[rateFunc, initial, 40, 3000];

ListPlot[data[[ ;; , 1 ;; 2]], Joined -> True, PlotMarkers -> None, PlotRange -> Full];
```