Physics through Computational Thinking

The Monte Carlo method

Auditya Sharma and Ambar Jain
Dept. of Physics, IISER Bhopal

Outline

In this module we look at

1. how to estimate ln(2) using the Monte Carlo method and a suitable definite integral.
2. how to estimate π using a Monte Carlo method to compute a suitable definite integral.

An elementary integral

Consider the integral

$$I = \int_0^1 \frac{1}{1+x}\, dx. \tag{1}$$

It is straightforward for us to find this integral. It is simply given by:

$$I = \ln(2). \tag{2}$$

In fact, we can get *Mathematica* to evaluate it for us:

```
Integrate[1/(1 + x), {x, 0, 1}]
```

```
Log[2]
```

Let us use the Monte Carlo method to estimate this integral, and therefore obtain an estimate for ln(2). The idea is to do a random sampling of points between zero and unity, and simply find an average of the value of integrand evaluated at these points.

```
data = Table[n = 2^k;
    {n, Mean[Table[Mean[Table[1/(RandomReal[] + 1), {n}]], {1000}]]}, {k, 5, 10}];
TableForm[data]
```

32	0.693579
64	0.692715
128	0.69312
256	0.693068
512	0.692981
1024	0.693385

```
ListPlot[data, Joined → True]
```




```
errdata = Table[n = 2^k;
    {n, Mean[Table[Abs[Mean[Table[1/(RandomReal[] + 1), {n}] - Log[2]]], {1000}]]}, {k, 5, 10}];
TableForm[errdata]
```

32	0.0192375
64	0.0144228
128	0.00955218
256	0.00683573
512	0.00479125
1024	0.00337105

```
fn[x_] = a x^b;

fit = FindFit[errdata, {a x^b}, {a, b}, {x}]
```

{a → 0.110299, b → -0.499775}

```
fn[x] /. fit;
```

```
Show[Plot[fn[x] /. fit, {x, 0, 1200}], ListPlot[errdata, Joined → True]]
```



So, the error goes roughly as $\frac{1}{\sqrt{n}}$.

Estimating π

Now, let us consider the integral

$$I = \int_0^1 \frac{4}{1+x^2}\,dx. \tag{3}$$

It is straightforward for us to find this integral. It is simply given by:

$$I = \pi. \tag{4}$$

We can get *Mathematica* to check it for us:

```
Integrate[4/(1 + x^2), {x, 0, 1}]
```

π

Let us use the Monte Carlo method to estimate this integral, and therefore obtain an estimate for π. The idea is to do a random sampling of points between zero and unity, and simply find an average of the value of integrand evaluated at these points, as before.

```
data = Table[n = 2^k;
    {n, Mean[Table[Mean[Table[4/((RandomReal[])^2 + 1), {n}]], {1000}]]}, {k, 5, 10}];
TableForm[data]
```

32	3.14287
64	3.1436
128	3.1434
256	3.13946
512	3.14373
1024	3.14179

```
ListPlot[data, Joined → True]
```




```
errdata = Table[n = 2^k;
    {n, Mean[Table[Abs[Mean[Table[4/((RandomReal[])^2 + 1), {n}] - π]], {100}]]}, {k, 5, 10}];
```

```
TableForm[errdata]
```

32	0.0941386
64	0.0631196
128	0.0487795
256	0.0337932
512	0.025524
1024	0.0145453

```
fn[x_] = a x^b;
```

```
fit = FindFit[errdata, {a x^b}, {a, b}, {x}]
```

{a → 0.515859, b → -0.493919}

```
fn[x] /. fit;
```

```
Show[Plot[fn[x] /. fit, {x, 0, 1200}], ListPlot[errdata, Joined → True]]
```



So, once again, we see that the error goes roughly as $\frac{1}{\sqrt{n}}$.